
1. [bookmark: _Toc122434485][bookmark: _Toc120711888]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk43 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc120711889]Corrections required
1. [bookmark: _Toc54888063][bookmark: _Toc100131355][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	f_TC_7_1_1_2_2_NR_TestBody ()

	Reason for change
	For this test case PDSCH aggregation factor 4 is used, which means DL PDSCH is expected to be transmitted on 4 consecutive slots available for transmission. For NR5GC operation, due to collision with scheduled systemInformation transmissions, the SS might end up transmitting the PDSCH on non-consecutive subframe/slots.

	Summary of change
	To avoid conflict with SystemInformation transmission which is scheduled in every even numbered frame, restrict the PDSCH transmission at step 5 & 9 to odd numbered frames, so that the DL PDSCH data can be transmitted on consecutive slots and verification at step 6 & 10 shall have the correct parameters.


	TTCN module
	MAC_TC_Common_NR.ttcn

	MCC160 Comment
	


[bookmark: _Toc54888064]
Before Change:
	function f_TC_7_1_1_2_2_NR_TestBody( DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC
  { //Correct Handling of DL HARQ process PDSCH Aggregation
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var integer v_Nack_Count := 0;
    var integer v_Ack_Count := 0;
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(nr_Cell1);
    var NR_SYSTEM_IND v_SYS_IND;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell1);
    var template (value) NR_DciDlInfo_Type v_DciDlInfo := f_NR_DciDlInfo_Init(v_NR_PhysicalParameters, cds_NR_DciFormat_1_1_Params_SetPDSCHarqIndicator(v_NR_PhysicalParameters.SubcarrierSpacing,cs_NR_DciFormat_1_X_PdschHarqTimingIndicator('011'B)), cs_NR_RedundancyVersionList_Def);
    var NR_MAC_SDU_Type v_EncodedRlcPdu;
    var template (value) NR_MAC_SDU_SubPDU_List_Type v_SDU_SubPDUListDL;
    var template (value) NR_MAC_PDUList_Type v_MAC_PDUListDL;

    timer t_Watchdog := 0.4;
    
    //Configure DL DCI as per Table 7.1.1.2.2.3.3-3: Physical layer parameters for DCI format 1_1 (Steps 3, 7, Table 7.1.1.2.2.3.2-1)
    //Set K1=5
    f_NR_SS_CommonCellConfig(nr_Cell1, cas_NR_DciDlInfo_REQ(nr_Cell1, cs_TimingInfo_Now, v_DciDlInfo));
    
    //By default in TS 38.508-1, the Time Alignment timer is set to infinity
    //configure SS for no Auto Time alignement - no grant needed
    f_NR_ULGrantConfiguration_Stop(nr_Cell1);   //@sic R5s221035 sic@
   
    //Config SS to report: NR_SystemRequest_Type -> NR_System_IndicationControl_Type -> UL_HARQ
    //NR_SYSTEM_IND -> NR_SystemIndication_Type-> UL_HARQ: ack/nack
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(enable));
    
    // PDCP SDU =36B +3B PDCP Headed+ 3B RLC Header + 2B MAC Header=44B =352 bits (an allowed TB size as per 38.523-3 Annex B)
    //@sic R5s221035 sic@
    v_EncodedRlcPdu := f_NR_RLC_AMD_FullPDU_Encvalue(0, tsc_NR_P_NoPoll, f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B(0, crs_NR_PDCP_SDU_36B)));
    v_SDU_SubPDUListDL := { cs_NR_MAC_SDU_SubPDU_LI8(tsc_LCID_SCG_Bearer, v_EncodedRlcPdu) };
    v_MAC_PDUListDL := { cs_NR_DL_MAC_PDU_SDU_SubPDU_Padding(v_SDU_SubPDUListDL) };
    
    // @siclog "Steps 3-4" siclog@
    // The SS transmits a downlink assignment addressed to the C-RNTI assigned to the UE, the rv_idx is 0.
    // The SS transmits in the indicated downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC error on UE side.
    f_NR_SS_L1_TestModeConfig(nr_Cell1, cas_NR_L1_TestMode_DL_SCH_CRC_REQ(nr_Cell1, Erroneous));
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 200);
    v_TimingUL := v_TimingDL;
    
    //Assign DL and UL timing
    select(v_SCS) {
        case(kHz15) {
            v_TimingDL.Subframe.Number := 1;
            v_TimingUL.Subframe.Number := 9;
        }
        case(kHz30) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 4;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case(kHz120) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology3 := 1;
            v_TimingUL.Subframe.Number := 1;
            v_TimingUL.Slot.SlotOffset.Numerology3 := 1;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SubCarrierSapcing"); }
    }
    t_Watchdog.start;
    DRB.send(cas_NR_DRB_COMMON_REQ_MAC_PDUList(nr_Cell1, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_MAC_PDUListDL));  //@sic R5s221035 sic@

    // @siclog "Step 5" siclog@
    // In the following 3 consecutive slots, the SS transmits on the same downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC error on UE side. (Note3)
    // SS already configured to do it - PDSCH aggregation
    
    // @siclog "Step 6" siclog@
    // Check: Does the UE transmit a HARQ NACK on slot n3+k1? (Note 4)
    alt {
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, nack))-> value v_SYS_IND
          {
            v_Nack_Count := v_Nack_Count + 1;
            repeat;
          }
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
            t_Watchdog.stop;
          }
        [] t_Watchdog.timeout {}
    }
    if (match(v_SYS_IND.Common.TimingInfo.SubFrame, v_TimingUL) and v_Nack_Count == 1) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    
    // @siclog "Step 7" siclog@
    // The SS transmits a downlink assignment addressed to the C-RNTI assigned to the UE, the rv_idx is 0.
    // @siclog "Step 8" siclog@
    // The SS transmits in the indicated downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC pass on UE side.
    //Set CRC error as normal mode
    f_NR_SS_L1_TestModeConfig(nr_Cell1, cas_NR_L1_TestMode_DL_SCH_CRC_REQ(nr_Cell1, Normal,cs_TimingInfo_Now));
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 200);
    v_TimingUL := v_TimingDL;
    
    //Assign DL and UL timing
    select(v_SCS) {
        case(kHz15) {
            v_TimingDL.Subframe.Number := 1;
            v_TimingUL.Subframe.Number := 9;
        }
        case(kHz30) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 4;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case(kHz120) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 1;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SubCarrierSapcing"); }
    }

    // @siclog "Step 9" siclog@
    // In the following 3 consecutive slots, the SS transmits on the same downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC pass on UE side. (Note3)
    t_Watchdog.start;
    DRB.send(cas_NR_DRB_COMMON_REQ_MAC_PDUList(nr_Cell1, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_MAC_PDUListDL));  //@sic R5s221035 sic@

    // @siclog "Step 10" siclog@
    // Check: Does the UE transmit a HARQ ACK on slot n3+k1? (Note 4)
    alt {
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack))-> value v_SYS_IND
          {
            v_Ack_Count := v_Ack_Count + 1;
            repeat;
          }
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, nack))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
            t_Watchdog.stop;
          }
        [] t_Watchdog.timeout {}
    }
    if (match(v_SYS_IND.Common.TimingInfo.SubFrame, v_TimingUL) and v_Ack_Count == 1) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
    }
    
    // Configure SS to stop indication of UL HARQ ACK/NACK
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));
    //@sic R5s221035 sic@
  }




After Change:
	function f_TC_7_1_1_2_2_NR_TestBody( DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC
  { //Correct Handling of DL HARQ process PDSCH Aggregation
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var integer v_Nack_Count := 0;
    var integer v_Ack_Count := 0;
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(nr_Cell1);
    var NR_SYSTEM_IND v_SYS_IND;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell1);
    var template (value) NR_DciDlInfo_Type v_DciDlInfo := f_NR_DciDlInfo_Init(v_NR_PhysicalParameters, cds_NR_DciFormat_1_1_Params_SetPDSCHarqIndicator(v_NR_PhysicalParameters.SubcarrierSpacing,cs_NR_DciFormat_1_X_PdschHarqTimingIndicator('011'B)), cs_NR_RedundancyVersionList_Def);
    var NR_MAC_SDU_Type v_EncodedRlcPdu;
    var template (value) NR_MAC_SDU_SubPDU_List_Type v_SDU_SubPDUListDL;
    var template (value) NR_MAC_PDUList_Type v_MAC_PDUListDL;

    timer t_Watchdog := 0.4;
    
    //Configure DL DCI as per Table 7.1.1.2.2.3.3-3: Physical layer parameters for DCI format 1_1 (Steps 3, 7, Table 7.1.1.2.2.3.2-1)
    //Set K1=5
    f_NR_SS_CommonCellConfig(nr_Cell1, cas_NR_DciDlInfo_REQ(nr_Cell1, cs_TimingInfo_Now, v_DciDlInfo));
    
    //By default in TS 38.508-1, the Time Alignment timer is set to infinity
    //configure SS for no Auto Time alignement - no grant needed
    f_NR_ULGrantConfiguration_Stop(nr_Cell1);   //@sic R5s221035 sic@
   
    //Config SS to report: NR_SystemRequest_Type -> NR_System_IndicationControl_Type -> UL_HARQ
    //NR_SYSTEM_IND -> NR_SystemIndication_Type-> UL_HARQ: ack/nack
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(enable));
    
    // PDCP SDU =36B +3B PDCP Headed+ 3B RLC Header + 2B MAC Header=44B =352 bits (an allowed TB size as per 38.523-3 Annex B)
    //@sic R5s221035 sic@
    v_EncodedRlcPdu := f_NR_RLC_AMD_FullPDU_Encvalue(0, tsc_NR_P_NoPoll, f_NR_PDCP_PDU_Encvalue(cs_NR_PDCP_PDU_18B(0, crs_NR_PDCP_SDU_36B)));
    v_SDU_SubPDUListDL := { cs_NR_MAC_SDU_SubPDU_LI8(tsc_LCID_SCG_Bearer, v_EncodedRlcPdu) };
    v_MAC_PDUListDL := { cs_NR_DL_MAC_PDU_SDU_SubPDU_Padding(v_SDU_SubPDUListDL) };
    
    // @siclog "Steps 3-4" siclog@
    // The SS transmits a downlink assignment addressed to the C-RNTI assigned to the UE, the rv_idx is 0.
    // The SS transmits in the indicated downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC error on UE side.
    f_NR_SS_L1_TestModeConfig(nr_Cell1, cas_NR_L1_TestMode_DL_SCH_CRC_REQ(nr_Cell1, Erroneous));
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 200);
    if (v_TimingDL.SFN.Number mod 2 == 0)
    {
      v_TimingDL.SFN.Number := (v_TimingDL.SFN.Number + 1) mod 1024;        
    }
    v_TimingUL := v_TimingDL;
    
    //Assign DL and UL timing
    select(v_SCS) {
        case(kHz15) {
            v_TimingDL.Subframe.Number := 1;
            v_TimingUL.Subframe.Number := 9;
        }
        case(kHz30) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 4;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case(kHz120) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology3 := 1;
            v_TimingUL.Subframe.Number := 1;
            v_TimingUL.Slot.SlotOffset.Numerology3 := 1;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SubCarrierSapcing"); }
    }
    t_Watchdog.start;
    DRB.send(cas_NR_DRB_COMMON_REQ_MAC_PDUList(nr_Cell1, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_MAC_PDUListDL));  //@sic R5s221035 sic@

    // @siclog "Step 5" siclog@
    // In the following 3 consecutive slots, the SS transmits on the same downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC error on UE side. (Note3)
    // SS already configured to do it - PDSCH aggregation
    
    // @siclog "Step 6" siclog@
    // Check: Does the UE transmit a HARQ NACK on slot n3+k1? (Note 4)
    alt {
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, nack))-> value v_SYS_IND
          {
            v_Nack_Count := v_Nack_Count + 1;
            repeat;
          }
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
            t_Watchdog.stop;
          }
        [] t_Watchdog.timeout {}
    }
    if (match(v_SYS_IND.Common.TimingInfo.SubFrame, v_TimingUL) and v_Nack_Count == 1) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
    }
    
    // @siclog "Step 7" siclog@
    // The SS transmits a downlink assignment addressed to the C-RNTI assigned to the UE, the rv_idx is 0.
    // @siclog "Step 8" siclog@
    // The SS transmits in the indicated downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC pass on UE side.
    //Set CRC error as normal mode
    f_NR_SS_L1_TestModeConfig(nr_Cell1, cas_NR_L1_TestMode_DL_SCH_CRC_REQ(nr_Cell1, Normal,cs_TimingInfo_Now));
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1, 200);
    if (v_TimingDL.SFN.Number mod 2 == 0)
    {
      v_TimingDL.SFN.Number := (v_TimingDL.SFN.Number + 1) mod 1024;        
    }
    v_TimingUL := v_TimingDL;
    
    //Assign DL and UL timing
    select(v_SCS) {
        case(kHz15) {
            v_TimingDL.Subframe.Number := 1;
            v_TimingUL.Subframe.Number := 9;
        }
        case(kHz30) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 4;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case(kHz120) {
            v_TimingDL.Subframe.Number := 0;
            v_TimingDL.Slot.SlotOffset.Numerology1 := 1;
            v_TimingUL.Subframe.Number := 1;
            v_TimingUL.Slot.SlotOffset.Numerology1 := 1;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SubCarrierSapcing"); }
    }

    // @siclog "Step 9" siclog@
    // In the following 3 consecutive slots, the SS transmits on the same downlink assignment a MAC PDU including a RLC PDU, The CRC is calculated in such a way, it will result in CRC pass on UE side. (Note3)
    t_Watchdog.start;
    DRB.send(cas_NR_DRB_COMMON_REQ_MAC_PDUList(nr_Cell1, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), v_MAC_PDUListDL));  //@sic R5s221035 sic@

    // @siclog "Step 10" siclog@
    // Check: Does the UE transmit a HARQ ACK on slot n3+k1? (Note 4)
    alt {
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, ack))-> value v_SYS_IND
          {
            v_Ack_Count := v_Ack_Count + 1;
            repeat;
          }
        [] SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(nr_Cell1, nack))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
            t_Watchdog.stop;
          }
        [] t_Watchdog.timeout {}
    }
    if (match(v_SYS_IND.Common.TimingInfo.SubFrame, v_TimingUL) and v_Ack_Count == 1) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 10");
    }
    
    // Configure SS to stop indication of UL HARQ ACK/NACK
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_UL_HARQ(disable));
    //@sic R5s221035 sic@
  }




